
Overview

This document lists TTCN changes needed to correct issues in the ATS ‘iWD-EUTRA-B2011-03_D12wk09’.

Corrections 

Change 1 – Correction to ‘fl_EUTRA_RRC_Procedure_Latency()’ and ‘f_TC_8_2_1_5_EUTRA()’

	Function name
	fl_EUTRA_RRC_Procedure_Latency ()   +
f_TC_8_2_1_5_EUTRA()

	Reason for change
	1. Race condition in function ‘fl_EUTRA_RRC_Procedure_Latency()’: 

When UE receives the indication of the first HARQ ACK, TTCN disables HARQ indications setting the confirmation flag to TRUE. It may happen that when TTCN is expecting the confirmation of the HARQ indication mode, the RRC response (SecurityModeComplete, step 6) is received, since these two messages are expected / received on different ports, and there is no warranty on the order. As a consequence the TC fails.


2. ‘fl_EUTRA_RRC_Procedure_Latency()’ and ‘f_TC_8_2_1_5_EUTRA()’ – missing TA command between RRC connection setup (step 4) and reconfiguration (step 7):

Function ‘fl_EUTRA_RRC_Procedure_Latency()’ always switches off the UL sync (TA sending), but the TA timer on RRC level is only set to infinity at the RRC connection reconfiguration (step 7) and the UE needs at least 1 TA to restart the TA timer, but it is not even started at step 3 / 4. 


	Summary of change
	1. Modified TTCN to disable L1 MAC indications without need for confirmation. 


2. Activated TA command after sending RRCConnectionSetup to set TA to infinity on UE side. Making sure that TA expires by waiting for 660 ms and then disable TA command.


	TTCN module
	8_2\RRC_ConnReconfig.ttcn

	MCC160 Comment
	


Before change

	...

  //----------------------------------------------------------------------------
  // Test case 8.2.1.5 and 8.2.1.6
  //----------------------------------------------------------------------------
  template SYSTEM_IND car_HarqError_IND(CellId_Type p_CellId) :=
  { /* @status    APPROVED */
    Common        := cr_IndAspCommonPart_SysInd (p_CellId, ?),
    Indication    := {
      HarqError := ?
    }
  };
  /*
   * @desc      send DL RRC PDU(s), receive corresponding UL RRC PDU(s) and check latency requirement given by p_N;
   *            there is one RRC procedure (8.2.1.5) or two (8.2.1.6); in the latter case there are 2 ASP to be sent (having the same timing info) and 2 ASP to be received
   *            HARQ retransmisions are compensated for the DL but will cause inconc for UL.
   * @param     p_CellId
   * @param     p_SRB_COMMON_REQ_1  .. ASP to be sent containing valid timing information and DL RRC PDU
   * @param     p_SRB_COMMON_REQ_2     (default value: omit; only present if there is a second parallel RRC procedure as in 8.2.1.6)
   * @param     p_SRB_COMMON_IND_1  .. ASP with expected UL RRC PDU and optionally NAS PDU
   * @param     p_SRB_COMMON_IND_2     (default value: omit; only present if there is a second parallel RRC procedure as in 8.2.1.6)
   * @param     p_IsFDD
   * @param     p_N                 .. N acc. to e.g. 36.523-1 cl. 8.2.1.6.2; N = 10, 15, 20 i.e. multiple of 5
   * @param     p_TestcaseStep
   * @return    SRB_COMMON_IND      .. latest IND-ASP
   * @status    APPROVED
   */
  function fl_EUTRA_RRC_Procedure_Latency(CellId_Type p_CellId,
                                          template (value)   SRB_COMMON_REQ p_SRB_COMMON_REQ_1,
                                          template (omit)    SRB_COMMON_REQ p_SRB_COMMON_REQ_2 := omit,
                                          template (present) SRB_COMMON_IND p_SRB_COMMON_IND_1,
                                          template           SRB_COMMON_IND p_SRB_COMMON_IND_2 := omit,
                                          boolean p_IsFDD,
                                          integer p_N,
                                          charstring p_TestcaseStep) runs on EUTRA_PTC return SRB_COMMON_IND
  {
    var integer v_DL_RetransmissionCnt := 0;
    var SYSTEM_IND v_SYSTEM_IND;
    var SRB_COMMON_IND v_SRB_COMMON_IND;
    var boolean v_T1Flag := false;
    var boolean v_T3Flag := false;
    var boolean v_T4Flag := (not isvalue(p_SRB_COMMON_REQ_2));     // we only check for a second UL PDU when there is a second DL PDU
    var integer v_NoOfGrants := 5;
    var integer v_GrantCycle;
    var integer v_DelayGrantToData;
    var SubFrameTiming_Type v_T0; // Sending of DL RRC PDU(s)
    var SubFrameTiming_Type v_T1; // HARQ ACK for DL RRC PDU(s)
    var SubFrameTiming_Type v_T2; // First UL grant
    var SubFrameTiming_Type v_T3; // First UL RRC PDU received
    var SubFrameTiming_Type v_T4; // Second UL RRC PDU received
    var SubFrameTiming_Type v_T5; // Last UL grant
    var SubFrameTiming_Type v_Tnow;
    var TimingInfo_Type v_TimingInfo;
    timer t_Timer;
    if (p_IsFDD) {
      v_GrantCycle := 8;          // HARQ RTT = 8
      v_DelayGrantToData := 4;    // UL data is always sent 4 TTIs after the corresponding grant
    } else {
      v_GrantCycle := 10;         // HARQ RTT = 10 for SN = 4
      v_DelayGrantToData := 7;    // UL data is sent 4..7 TTIs after the corresponding grant
    }
    //Get timing reference for calculating expected HARQ ACK i.e. when the performance measurement start
    v_TimingInfo := valueof(p_SRB_COMMON_REQ_1.Common.TimingInfo);
    v_T0 := v_TimingInfo.SubFrame; /* T0: DL PDU is sent the first time; subframe number shall be 4
                                    *     => possible retransmissions are at
                                    *        T0 + 8,  T0 + 16, T0 + 24, T0 + 32    for FDD
                                    *        T0 + 10, T0 + 20, T0 + 30, T0 + 40    for TDD */
    v_T2 := f_EUTRA_TimingInfoAdd(v_T0, p_N);                       /* T2: first UL grant is assigned; sub-sequent UL grants are to cope with DL retransmissions */
    f_SS_ConfigUL_HARQ_IndMode(p_CellId, tsc_L1Mac_IndicationMode_Enable, v_TimingInfo);         /* enable indication of HARQ ack/nack
                                                                                                    @sic R5s120024: scheduled configuration sic@ */
    f_EUTRA_ULGrantConfiguration_Common(p_CellId,
                                        cs_TimingInfo(v_T2.SFN.Number, v_T2.Subframe.Number),
                                        cs_DciInfo_CcchDcchDtchUL_Explicit(4, 15), //@sic R5s120099r1 sic@
                                        cs_PUCCH_Synch_None,
                                        cs_UL_GrantConfig_NoOfGrants(v_GrantCycle, v_NoOfGrants));
    v_T5 := f_EUTRA_TimingInfoAdd(v_T0, v_GrantCycle * (v_NoOfGrants - 1));                    /* T5: last UL grant is sent out */
    v_Tnow := f_EUTRA_GetCurrentTiming(p_CellId);
    t_Timer.start (int2float(f_EUTRA_SubFrameTimingDuration(v_Tnow, v_T5) + 50) / 1000.0);     /* 50ms added to be at the safe side */
    SRB.send(p_SRB_COMMON_REQ_1);                                                              /* send (first) DL PDU at T0 */
    if (isvalue(p_SRB_COMMON_REQ_2)) {
      SRB.send(p_SRB_COMMON_REQ_2);                                                            /* send second DL PDU at T0 */
    }
    alt {
      [] SYSIND.receive (car_UL_HARQ_IND(p_CellId, ?, nack)) {                                                 /* HARQ NACK */
        v_DL_RetransmissionCnt := v_DL_RetransmissionCnt + 1;
        repeat;
      }
      [] SYSIND.receive (car_HarqError_IND(p_CellId)) -> value v_SYSTEM_IND {                                  /* HARQ error indication */
        if (ischosen(v_SYSTEM_IND.Indication.HarqError.UL)) {
          f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, p_TestcaseStep & "HARQ error for UL transmission in latency test");
        } else {
          repeat;
        }
      }
      [not v_T1Flag] SYSIND.receive (car_UL_HARQ_IND(p_CellId, ?, ack)) -> value v_SYSTEM_IND {                /* HARQ ACK => T1 */
        v_T1 := v_SYSTEM_IND.Common.TimingInfo.SubFrame;
        f_SS_ConfigUL_HARQ_IndMode(p_CellId, tsc_L1Mac_IndicationMode_Disable);                                /* disable indication of HARQ ack/nack */
        v_T1Flag := true;
        repeat;
      }
      [not v_T3Flag] SRB.receive(p_SRB_COMMON_IND_1) -> value v_SRB_COMMON_IND {                               /* First UL RRC PDU => T3 */
        v_T3 := v_SRB_COMMON_IND.Common.TimingInfo.SubFrame;
        v_T3Flag := true;
        repeat;      // => wait for timeout to be sure that there are no UL grants pending anymore
      }
      [not v_T4Flag] SRB.receive(p_SRB_COMMON_IND_2) -> value v_SRB_COMMON_IND {                               /* Second UL RRC PDU => T4 */
        if (not v_T3Flag) {
          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep & "Incorrect order of UL PDUs");
        }
        v_T4 := v_SRB_COMMON_IND.Common.TimingInfo.SubFrame;
        v_T4Flag := true;
        repeat;      // => wait for timeout to be sure that there are no UL grants pending anymore
      }
      [] t_Timer.timeout {                                                                                     /* timeout => after T5 */
        // exit the alt
      }
    }
    if (not v_T1Flag) {
      f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, p_TestcaseStep & "no HARQ ACK for DL PDU");
    }
    if (not v_T3Flag) {
      f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep & "no UL PDU");
    }
    if (not v_T4Flag) {
      f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep & "no second UL PDU");
    }
    // Latency check:
    if ((f_EUTRA_SubFrameTimingDuration(v_T1, v_T3) > p_N + v_DelayGrantToData) or                                     /* check first RRC procedure regarding latency requirenments */
        (isvalue(p_SRB_COMMON_REQ_2) and (f_EUTRA_SubFrameTimingDuration(v_T1, v_T4) > p_N + v_DelayGrantToData))) {   /* check second RRC procedure regarding latency requirenments */
      f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep);
    } else {
      f_EUTRA_PreliminaryPass(__FILE__, __LINE__, p_TestcaseStep);
    }
    return v_SRB_COMMON_IND;
  }
  /*
   * @desc      REFERENCE TS 36.523-1 clause 8.2.1.5
   * @status    APPROVED
   */
  function f_TC_8_2_1_5_EUTRA() runs on EUTRA_PTC
  { /* RRC connection reconfiguration / Radio bearer establishment for transition from RRC_Idle to RRC CONNECTED / Success / Latency check */
    var EUTRA_FDD_TDD_CellInfo_Type v_EUTRA_FDD_TDD_Info;
    var EUTRA_FDD_TDD_Mode_Type v_EUTRA_FDD_TDD_Mode;
    var PhysicalConfigDedicated_AntennaInfo_Type v_AntennaInfo;
    var EUTRA_SecurityParams_Type v_Auth_Params;
    var O4_Type v_NasCountUL;
    var NAS_KsiValue v_KsiValue;
    var SRB_COMMON_IND v_ReceivedAsp;
    var SubFrameTiming_Type v_Timing;
    var integer v_N;
    var boolean v_IsFDD;
    f_EUTRA_Init(c1);
    //Get FDD or TDD mode from Cell-configuration
    v_EUTRA_FDD_TDD_Info := f_EUTRA_CellInfo_GetFDD_TDD_Info (eutra_Cell1);
    v_EUTRA_FDD_TDD_Mode := f_EUTRA_CellInfo_GetFDD_TDD (eutra_Cell1);
    v_AntennaInfo := f_EUTRA_CellInfo_GetAntennaInfoDedicated (eutra_Cell1);
    v_IsFDD := (v_EUTRA_FDD_TDD_Mode == FDD);
    //Create and configure cell
    f_EUTRA_CellConfig_Def(eutra_Cell1);
    /* Preamble to enter UE in E-UTRA RRC_IDLE (state 2) */
    f_EUTRA_Preamble (eutra_Cell1, STATE2_IDLEUPDATE);
    //Get security parameters
    v_Auth_Params := f_EUTRA_Security_Get();
    v_KsiValue := f_EUTRA_SecurityKSIasme_Get();
    // enable reporting of HARQ errors
    f_EUTRA_SS_CommonL1MacIndCtrlConfig(eutra_Cell1, cas_HarqErrorIndication_REQ(eutra_Cell1));
    f_EUTRA_TestBody_Set(true);
    //@siclog "Step 1" siclog@
    //The SS sends a Paging message to the UE on the appropriate paging block, and including the UE identity in one entry of the IE pagingRecordLists.
    f_EUTRA_UE_Page_Def(eutra_Cell1);
    //@siclog "Step 2" siclog@
    //The UE transmits an RRCConnectionRequest message.
    f_EUTRA_RRC_ConnectionRequest_Def(eutra_Cell1, cr_EstablishmentCause_mt_Access);    // @sic R5-120721 change 3 sic@
    //@siclog "Step 3" siclog@
    //The SS transmit an RRCConnectionSetup message. (Note 3)
    //@siclog "Step 4" siclog@
    //Check: Does the UE transmit an RRCConnectionSetupComplete message 19 subframes after successful completion of step 3? (Note 2)
    //Note: to avoid unexpected TA commands (resulting in additional HARQ indications) the time alignment timer needs to be set to infinitiv
    v_N := 15;
    v_Timing := f_EUTRA_GetNextSendOccasion(eutra_Cell1);
    v_ReceivedAsp := fl_EUTRA_RRC_Procedure_Latency(eutra_Cell1,
                                                    cas_SRB0_RrcPdu_REQ(eutra_Cell1,
                                                                        cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),
                                                                        f_EUTRA_508_RRCConnectionSetup(eutra_Cell1, -, infinity_)),   /* @sic R5-120721 change 3 sic@ */
                                                    -,
                                                    car_SRB1_RrcNasPdu_IND(eutra_Cell1,
                                                                           cr_508_RRCConnectionSetupComplete(tsc_RRC_TI_Def, ?),
                                                                           cr_NAS_Indication(tsc_SHT_IntegrityProtected,
                                                                                             cr_508_SERVICE_REQUEST(v_KsiValue))),
                                                    -,
                                                    v_IsFDD,
                                                    v_N,
                                                    "Test Case 8.2.1.5 Step 4");
    v_NasCountUL := v_ReceivedAsp.Signalling.Nas[0].SecurityProtection.NasCount;
    v_Auth_Params := f_EUTRA_Authentication_InitAS(v_Auth_Params, v_NasCountUL);        /* Calculate Kenb and KRRCenc, KRRCint, KUPenc */
    // Read PDCP SQN for all RBs and calcuate Ciphering Activation times
    v_Auth_Params.AS_Ciphering.ActTimeList := f_EUTRA_RRC_CipherActTime_Get(eutra_Cell1);
    f_EUTRA_SS_RRC_EnableIntegrityProtection(eutra_Cell1, v_Auth_Params.AS_Integrity);  /* SecurityModeCommand is integrity protected with the algorithm
                                                                                           being contained in the message; all sub-sequent RRC messages
                                                                                           are integrity protected in UL and DL (36.331 cl. 5.3.4.3)    */
    f_EUTRA_SS_RRC_EnableCipheringULandDL(eutra_Cell1, v_Auth_Params.AS_Ciphering);
    //@siclog "Step 5" siclog@
    //The SS transmits a SecurityModeCommand message to activate AS security. (Note 3)
    //@siclog "Step 6" siclog@
    //Check: Does the UE transmit a SecurityModeComplete message within 14 (FDD)/17 (TDD) subframes after successful completion of step 5? (Note 2)
    v_Timing := f_EUTRA_GetNextSendOccasion(eutra_Cell1);
    v_N := 10;
    fl_EUTRA_RRC_Procedure_Latency(eutra_Cell1,
                                   cas_SRB1_RrcPdu_REQ(eutra_Cell1,
                                                       cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),
                                                       cs_508_SecurityModeCommand(tsc_RRC_TI_Def,
                                                                                  v_Auth_Params.AS_Integrity.Algorithm,
                                                                                  v_Auth_Params.AS_Ciphering.Algorithm)),
                                   -,
                                   car_SRB1_RrcPdu_IND(eutra_Cell1,
                                                       cr_508_SecurityModeComplete(tsc_RRC_TI_Def)),
                                   -,
                                   v_IsFDD,
                                   v_N,
                                   "Test Case 8.2.1.5 Step 6");
    //@siclog "Step 7" siclog@
    //The SS transmits an RRCConnectionReconfiguration message to establish a data radio bearer. (Note 3)
    //@siclog "Step 8" siclog@
    //Check: Does the UE transmit an RRCConnectionReconfigurationComplete message 19 subframes after successful completion of step 7? (Note 2)
    v_Timing := f_EUTRA_GetNextSendOccasion ( eutra_Cell1 );
    v_N := 15;
    fl_EUTRA_RRC_Procedure_Latency(eutra_Cell1,
                                   cas_SRB1_RrcPdu_REQ(eutra_Cell1,
                                                       cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),
                                                       cs_RRCConnectionReconfiguration_RadioResConfig(tsc_RRC_TI_Def,
                                                                                                      cs_508_RadioResourceConfigDedicated_Srb2({ cs_508_DRB_ToAddMod_DEFAULT_AM(tsc_DRB1) },
                                                                                                                                               cs_MAC_MainConfig_Explicit_TA_Infinity,  /* @sic R5s110530 change1; R5-113706 sic@ */
                                                                                                                                               cs_508_PhysicalConfigDedicated_Default_RBC(v_EUTRA_FDD_TDD_Info.cqi_ReportConfig,
                                                                                                                                                                                          v_EUTRA_FDD_TDD_Info.soundingRS_UL_Config,
                                                                                                                                                                                          v_AntennaInfo)))),
                                   -,
                                   car_SRB1_RrcPdu_IND(eutra_Cell1,
                                                       cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)),
                                   -,
                                   v_IsFDD,
                                   v_N,
                                   "Test Case 8.2.1.5 Step 8");
    //@siclog "Step 9" siclog@
    //The SS transmits a UECapabilityEnquiry message to request UE radio access capability information for E UTRA only. (Note 3)
    //@siclog "Step 10" siclog@
    //Check: Does the UE transmit a UECapabilityInformation message 14 subframes after successful completion of step 9? (Note 2)
    v_Timing := f_EUTRA_GetNextSendOccasion ( eutra_Cell1 );
    v_N := 10;
    fl_EUTRA_RRC_Procedure_Latency(eutra_Cell1,
                                   cas_SRB1_RrcPdu_REQ(eutra_Cell1,
                                                       cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),
                                                       cs_508_UeCapabilityEnquiry(tsc_RRC_TI_Def)),
                                   -,
                                   car_SRB1_RrcPdu_IND(eutra_Cell1,
                                                       cr_508_UeCapabilityInformation(tsc_RRC_TI_Def, cr_EutraCapInfo)),
                                   -,
                                   v_IsFDD,
                                   v_N,
                                   "Test Case 8.2.1.5 Step 10");
    f_EUTRA_TestBody_Set(false);
    /* Postamble */
    f_EUTRA_StartDefULGrantTransmission(eutra_Cell1);           // UL grants per scheduling request
    f_EUTRA_Postamble(eutra_Cell1, E2_CONNECTED);
  }
...


After change

	...

  //----------------------------------------------------------------------------
  // Test case 8.2.1.5 and 8.2.1.6
  //----------------------------------------------------------------------------
  template SYSTEM_IND car_HarqError_IND(CellId_Type p_CellId) :=
  { /* @status    APPROVED */
    Common        := cr_IndAspCommonPart_SysInd (p_CellId, ?),
    Indication    := {
      HarqError := ?
    }
  };
  /*
   * @desc      send DL RRC PDU(s), receive corresponding UL RRC PDU(s) and check latency requirement given by p_N;
   *            there is one RRC procedure (8.2.1.5) or two (8.2.1.6); in the latter case there are 2 ASP to be sent (having the same timing info) and 2 ASP to be received
   *            HARQ retransmisions are compensated for the DL but will cause inconc for UL.
   * @param     p_CellId
   * @param     p_SRB_COMMON_REQ_1  .. ASP to be sent containing valid timing information and DL RRC PDU
   * @param     p_SRB_COMMON_REQ_2     (default value: omit; only present if there is a second parallel RRC procedure as in 8.2.1.6)
   * @param     p_SRB_COMMON_IND_1  .. ASP with expected UL RRC PDU and optionally NAS PDU
   * @param     p_SRB_COMMON_IND_2     (default value: omit; only present if there is a second parallel RRC procedure as in 8.2.1.6)
   * @param     p_IsFDD
   * @param     p_N                 .. N acc. to e.g. 36.523-1 cl. 8.2.1.6.2; N = 10, 15, 20 i.e. multiple of 5
   * @param     p_TestcaseStep
   * @return    SRB_COMMON_IND      .. latest IND-ASP
   * @status    APPROVED
   */
  function fl_EUTRA_RRC_Procedure_Latency(CellId_Type p_CellId,
                                          template (value)   SRB_COMMON_REQ p_SRB_COMMON_REQ_1,
                                          template (omit)    SRB_COMMON_REQ p_SRB_COMMON_REQ_2 := omit,
                                          template (present) SRB_COMMON_IND p_SRB_COMMON_IND_1,
                                          template           SRB_COMMON_IND p_SRB_COMMON_IND_2 := omit,
                                          boolean p_IsFDD,
                                          integer p_N,
                                          charstring p_TestcaseStep) runs on EUTRA_PTC return SRB_COMMON_IND
  {
    var integer v_DL_RetransmissionCnt := 0;
    var SYSTEM_IND v_SYSTEM_IND;
    var SRB_COMMON_IND v_SRB_COMMON_IND;
    var boolean v_T1Flag := false;
    var boolean v_T3Flag := false;
    var boolean v_T4Flag := (not isvalue(p_SRB_COMMON_REQ_2));     // we only check for a second UL PDU when there is a second DL PDU
    var integer v_NoOfGrants := 5;
    var integer v_GrantCycle;
    var integer v_DelayGrantToData;
    var SubFrameTiming_Type v_T0; // Sending of DL RRC PDU(s)
    var SubFrameTiming_Type v_T1; // HARQ ACK for DL RRC PDU(s)
    var SubFrameTiming_Type v_T2; // First UL grant
    var SubFrameTiming_Type v_T3; // First UL RRC PDU received
    var SubFrameTiming_Type v_T4; // Second UL RRC PDU received
    var SubFrameTiming_Type v_T5; // Last UL grant
    var SubFrameTiming_Type v_Tnow;
    var TimingInfo_Type v_TimingInfo;
    var template (value) SYSTEM_CTRL_REQ v_SYSTEM_CTRL_REQ;
    timer t_Timer;
    if (p_IsFDD) {
      v_GrantCycle := 8;          // HARQ RTT = 8
      v_DelayGrantToData := 4;    // UL data is always sent 4 TTIs after the corresponding grant
    } else {
      v_GrantCycle := 10;         // HARQ RTT = 10 for SN = 4
      v_DelayGrantToData := 7;    // UL data is sent 4..7 TTIs after the corresponding grant
    }
    //Get timing reference for calculating expected HARQ ACK i.e. when the performance measurement start
    v_TimingInfo := valueof(p_SRB_COMMON_REQ_1.Common.TimingInfo);
    v_T0 := v_TimingInfo.SubFrame; /* T0: DL PDU is sent the first time; subframe number shall be 4
                                    *     => possible retransmissions are at
                                    *        T0 + 8,  T0 + 16, T0 + 24, T0 + 32    for FDD
                                    *        T0 + 10, T0 + 20, T0 + 30, T0 + 40    for TDD */
    v_T2 := f_EUTRA_TimingInfoAdd(v_T0, p_N);                       /* T2: first UL grant is assigned; sub-sequent UL grants are to cope with DL retransmissions */
    f_SS_ConfigUL_HARQ_IndMode(p_CellId, tsc_L1Mac_IndicationMode_Enable, v_TimingInfo);         /* enable indication of HARQ ack/nack
                                                                                                    @sic R5s120024: scheduled configuration sic@ */
    f_EUTRA_ULGrantConfiguration_Common(p_CellId,
                                        cs_TimingInfo(v_T2.SFN.Number, v_T2.Subframe.Number),
                                        cs_DciInfo_CcchDcchDtchUL_Explicit(4, 15), //@sic R5s120099r1 sic@
                                        cs_PUCCH_Synch_None,
                                        cs_UL_GrantConfig_NoOfGrants(v_GrantCycle, v_NoOfGrants));
    v_T5 := f_EUTRA_TimingInfoAdd(v_T0, v_GrantCycle * (v_NoOfGrants - 1));                    /* T5: last UL grant is sent out */
    v_Tnow := f_EUTRA_GetCurrentTiming(p_CellId);
    t_Timer.start (int2float(f_EUTRA_SubFrameTimingDuration(v_Tnow, v_T5) + 50) / 1000.0);     /* 50ms added to be at the safe side */
    SRB.send(p_SRB_COMMON_REQ_1);                                                              /* send (first) DL PDU at T0 */
    if (isvalue(p_SRB_COMMON_REQ_2)) {
      SRB.send(p_SRB_COMMON_REQ_2);                                                            /* send second DL PDU at T0 */
    }
    alt {
      [] SYSIND.receive (car_UL_HARQ_IND(p_CellId, ?, nack)) {                                                 /* HARQ NACK */
        v_DL_RetransmissionCnt := v_DL_RetransmissionCnt + 1;
        repeat;
      }
      [] SYSIND.receive (car_HarqError_IND(p_CellId)) -> value v_SYSTEM_IND {                                  /* HARQ error indication */
        if (ischosen(v_SYSTEM_IND.Indication.HarqError.UL)) {
          f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, p_TestcaseStep & "HARQ error for UL transmission in latency test");
        } else {
          repeat;
        }
      }
      [not v_T1Flag] SYSIND.receive (car_UL_HARQ_IND(p_CellId, ?, ack)) -> value v_SYSTEM_IND {                /* HARQ ACK => T1 */
        v_T1 := v_SYSTEM_IND.Common.TimingInfo.SubFrame;
        v_SYSTEM_CTRL_REQ := cas_UL_HARQ_IndConfig_REQ (p_CellId, tsc_L1Mac_IndicationMode_Disable);
        v_SYSTEM_CTRL_REQ.Common.ControlInfo.CnfFlag := false;
        f_EUTRA_SS_CommonL1MacIndCtrlConfig(p_CellId, v_SYSTEM_CTRL_REQ);
        v_T1Flag := true;
        repeat;
      }
      [not v_T3Flag] SRB.receive(p_SRB_COMMON_IND_1) -> value v_SRB_COMMON_IND {                               /* First UL RRC PDU => T3 */
        v_T3 := v_SRB_COMMON_IND.Common.TimingInfo.SubFrame;
        v_T3Flag := true;
        repeat;      // => wait for timeout to be sure that there are no UL grants pending anymore
      }
      [not v_T4Flag] SRB.receive(p_SRB_COMMON_IND_2) -> value v_SRB_COMMON_IND {                               /* Second UL RRC PDU => T4 */
        if (not v_T3Flag) {
          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep & "Incorrect order of UL PDUs");
        }
        v_T4 := v_SRB_COMMON_IND.Common.TimingInfo.SubFrame;
        v_T4Flag := true;
        repeat;      // => wait for timeout to be sure that there are no UL grants pending anymore
      }
      [] t_Timer.timeout {                                                                                     /* timeout => after T5 */
        // exit the alt
      }
    }
    if (not v_T1Flag) {
      f_EUTRA_SetVerdictInconc(__FILE__, __LINE__, p_TestcaseStep & "no HARQ ACK for DL PDU");
    }
    if (not v_T3Flag) {
      f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep & "no UL PDU");
    }
    if (not v_T4Flag) {
      f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep & "no second UL PDU");
    }
    // Latency check:
    if ((f_EUTRA_SubFrameTimingDuration(v_T1, v_T3) > p_N + v_DelayGrantToData) or                                     /* check first RRC procedure regarding latency requirenments */
        (isvalue(p_SRB_COMMON_REQ_2) and (f_EUTRA_SubFrameTimingDuration(v_T1, v_T4) > p_N + v_DelayGrantToData))) {   /* check second RRC procedure regarding latency requirenments */
      f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep);
    } else {
      f_EUTRA_PreliminaryPass(__FILE__, __LINE__, p_TestcaseStep);
    }
    f_EUTRA_SS_CommonCellConfig(p_CellId, cas_PUCCH_Synch_Config_REQ(p_CellId, cs_TimingInfo_Now, cs_PUCCH_Synch_Auto_Def));
    return v_SRB_COMMON_IND;
  }
  /*
   * @desc      REFERENCE TS 36.523-1 clause 8.2.1.5
   * @status    APPROVED
   */
  function f_TC_8_2_1_5_EUTRA() runs on EUTRA_PTC
  { /* RRC connection reconfiguration / Radio bearer establishment for transition from RRC_Idle to RRC CONNECTED / Success / Latency check */
    var EUTRA_FDD_TDD_CellInfo_Type v_EUTRA_FDD_TDD_Info;
    var EUTRA_FDD_TDD_Mode_Type v_EUTRA_FDD_TDD_Mode;
    var PhysicalConfigDedicated_AntennaInfo_Type v_AntennaInfo;
    var EUTRA_SecurityParams_Type v_Auth_Params;
    var O4_Type v_NasCountUL;
    var NAS_KsiValue v_KsiValue;
    var SRB_COMMON_IND v_ReceivedAsp;
    var SubFrameTiming_Type v_Timing;
    var integer v_N;
    var boolean v_IsFDD;
    f_EUTRA_Init(c1);
    //Get FDD or TDD mode from Cell-configuration
    v_EUTRA_FDD_TDD_Info := f_EUTRA_CellInfo_GetFDD_TDD_Info (eutra_Cell1);
    v_EUTRA_FDD_TDD_Mode := f_EUTRA_CellInfo_GetFDD_TDD (eutra_Cell1);
    v_AntennaInfo := f_EUTRA_CellInfo_GetAntennaInfoDedicated (eutra_Cell1);
    v_IsFDD := (v_EUTRA_FDD_TDD_Mode == FDD);
    //Create and configure cell
    f_EUTRA_CellConfig_Def(eutra_Cell1);
    /* Preamble to enter UE in E-UTRA RRC_IDLE (state 2) */
    f_EUTRA_Preamble (eutra_Cell1, STATE2_IDLEUPDATE);
    //Get security parameters
    v_Auth_Params := f_EUTRA_Security_Get();
    v_KsiValue := f_EUTRA_SecurityKSIasme_Get();
    // enable reporting of HARQ errors
    f_EUTRA_SS_CommonL1MacIndCtrlConfig(eutra_Cell1, cas_HarqErrorIndication_REQ(eutra_Cell1));
    f_EUTRA_TestBody_Set(true);
    //@siclog "Step 1" siclog@
    //The SS sends a Paging message to the UE on the appropriate paging block, and including the UE identity in one entry of the IE pagingRecordLists.
    f_EUTRA_UE_Page_Def(eutra_Cell1);
    //@siclog "Step 2" siclog@
    //The UE transmits an RRCConnectionRequest message.
    f_EUTRA_RRC_ConnectionRequest_Def(eutra_Cell1, cr_EstablishmentCause_mt_Access);    // @sic R5-120721 change 3 sic@
    //@siclog "Step 3" siclog@
    //The SS transmit an RRCConnectionSetup message. (Note 3)
    //@siclog "Step 4" siclog@
    //Check: Does the UE transmit an RRCConnectionSetupComplete message 19 subframes after successful completion of step 3? (Note 2)
    //Note: to avoid unexpected TA commands (resulting in additional HARQ indications) the time alignment timer needs to be set to infinitiv
    v_N := 15;
    v_Timing := f_EUTRA_GetNextSendOccasion(eutra_Cell1);
    v_ReceivedAsp := fl_EUTRA_RRC_Procedure_Latency(eutra_Cell1,
                                                    cas_SRB0_RrcPdu_REQ(eutra_Cell1,
                                                                        cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),
                                                                        f_EUTRA_508_RRCConnectionSetup(eutra_Cell1, -, infinity_)),   /* @sic R5-120721 change 3 sic@ */
                                                    -,
                                                    car_SRB1_RrcNasPdu_IND(eutra_Cell1,
                                                                           cr_508_RRCConnectionSetupComplete(tsc_RRC_TI_Def, ?),
                                                                           cr_NAS_Indication(tsc_SHT_IntegrityProtected,
                                                                                             cr_508_SERVICE_REQUEST(v_KsiValue))),
                                                    -,
                                                    v_IsFDD,
                                                    v_N,
                                                    "Test Case 8.2.1.5 Step 4");
    v_NasCountUL := v_ReceivedAsp.Signalling.Nas[0].SecurityProtection.NasCount;
    v_Auth_Params := f_EUTRA_Authentication_InitAS(v_Auth_Params, v_NasCountUL);        /* Calculate Kenb and KRRCenc, KRRCint, KUPenc */
    // Read PDCP SQN for all RBs and calcuate Ciphering Activation times
    v_Auth_Params.AS_Ciphering.ActTimeList := f_EUTRA_RRC_CipherActTime_Get(eutra_Cell1);
    f_EUTRA_SS_RRC_EnableIntegrityProtection(eutra_Cell1, v_Auth_Params.AS_Integrity);  /* SecurityModeCommand is integrity protected with the algorithm
                                                                                           being contained in the message; all sub-sequent RRC messages
                                                                                           are integrity protected in UL and DL (36.331 cl. 5.3.4.3)    */
    f_EUTRA_SS_RRC_EnableCipheringULandDL(eutra_Cell1, v_Auth_Params.AS_Ciphering);
    //@siclog "Step 5" siclog@
    //The SS transmits a SecurityModeCommand message to activate AS security. (Note 3)
    //@siclog "Step 6" siclog@
    //Check: Does the UE transmit a SecurityModeComplete message within 14 (FDD)/17 (TDD) subframes after successful completion of step 5? (Note 2)
    v_Timing := f_EUTRA_GetNextSendOccasion(eutra_Cell1);
    v_N := 10;
    fl_EUTRA_RRC_Procedure_Latency(eutra_Cell1,
                                   cas_SRB1_RrcPdu_REQ(eutra_Cell1,
                                                       cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),
                                                       cs_508_SecurityModeCommand(tsc_RRC_TI_Def,
                                                                                  v_Auth_Params.AS_Integrity.Algorithm,
                                                                                  v_Auth_Params.AS_Ciphering.Algorithm)),
                                   -,
                                   car_SRB1_RrcPdu_IND(eutra_Cell1,
                                                       cr_508_SecurityModeComplete(tsc_RRC_TI_Def)),
                                   -,
                                   v_IsFDD,
                                   v_N,
                                   "Test Case 8.2.1.5 Step 6");
    //@siclog "Step 7" siclog@
    //The SS transmits an RRCConnectionReconfiguration message to establish a data radio bearer. (Note 3)
    //@siclog "Step 8" siclog@
    //Check: Does the UE transmit an RRCConnectionReconfigurationComplete message 19 subframes after successful completion of step 7? (Note 2)
    v_Timing := f_EUTRA_GetNextSendOccasion ( eutra_Cell1 );
    v_N := 15;
    fl_EUTRA_RRC_Procedure_Latency(eutra_Cell1,
                                   cas_SRB1_RrcPdu_REQ(eutra_Cell1,
                                                       cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),
                                                       cs_RRCConnectionReconfiguration_RadioResConfig(tsc_RRC_TI_Def,
                                                                                                      cs_508_RadioResourceConfigDedicated_Srb2({ cs_508_DRB_ToAddMod_DEFAULT_AM(tsc_DRB1) },
                                                                                                                                               cs_MAC_MainConfig_Explicit_TA_Infinity,  /* @sic R5s110530 change1; R5-113706 sic@ */
                                                                                                                                               cs_508_PhysicalConfigDedicated_Default_RBC(v_EUTRA_FDD_TDD_Info.cqi_ReportConfig,
                                                                                                                                                                                          v_EUTRA_FDD_TDD_Info.soundingRS_UL_Config,
                                                                                                                                                                                          v_AntennaInfo)))),
                                   -,
                                   car_SRB1_RrcPdu_IND(eutra_Cell1,
                                                       cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)),
                                   -,
                                   v_IsFDD,
                                   v_N,
                                   "Test Case 8.2.1.5 Step 8");
    f_Delay(0.660);
    f_EUTRA_SS_CommonCellConfig(eutra_Cell1, cas_PUCCH_Synch_Config_REQ(eutra_Cell1, cs_TimingInfo_Now, cs_PUCCH_Synch_None));
    //@siclog "Step 9" siclog@
    //The SS transmits a UECapabilityEnquiry message to request UE radio access capability information for E UTRA only. (Note 3)
    //@siclog "Step 10" siclog@
    //Check: Does the UE transmit a UECapabilityInformation message 14 subframes after successful completion of step 9? (Note 2)
    v_Timing := f_EUTRA_GetNextSendOccasion ( eutra_Cell1 );
    v_N := 10;
    fl_EUTRA_RRC_Procedure_Latency(eutra_Cell1,
                                   cas_SRB1_RrcPdu_REQ(eutra_Cell1,
                                                       cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),
                                                       cs_508_UeCapabilityEnquiry(tsc_RRC_TI_Def)),
                                   -,
                                   car_SRB1_RrcPdu_IND(eutra_Cell1,
                                                       cr_508_UeCapabilityInformation(tsc_RRC_TI_Def, cr_EutraCapInfo)),
                                   -,
                                   v_IsFDD,
                                   v_N,
                                   "Test Case 8.2.1.5 Step 10");
    f_EUTRA_TestBody_Set(false);
    /* Postamble */
    f_EUTRA_StartDefULGrantTransmission(eutra_Cell1);           // UL grants per scheduling request
    f_EUTRA_Postamble(eutra_Cell1, E2_CONNECTED);
  }
...

