
Overview

This document lists TTCN changes needed to correct issues in the ATS ‘iWD-EUTRA-B2011-03_D12wk09’.

Corrections 

Change 1 – Correction to fl_EUTRA_DRB_IdentityList
	Test case name
	fl_EUTRA_DRB_IdentityList

	Reason for change
	In test step 4 of fl_EUTRA_7_2_3_21_TestBody, v_DRB_Config is set to cs_DRB1_ConfigAM_TM_Mode which sets DRB1 to TM mode before a call to f_EUTRA_508RRC_IntraLTE_HO_IntraCell_Step1_5 is done where v_DRB_Config is passed. 

In f_EUTRA_508RRC_IntraLTE_HO_IntraCell_Step1_5 there is a call to fl_EUTRA_DRB_IdentityList that returns a DRB_IdentityList_AllAndAMonly_Type assigned to v_DRB_IdentityList. v_DRB_IdentityList.AMonly is later on used on test step 5 and this produces an execution error due to a non initialized variable as there are no AM DRBs.

	Summary of change
	fl_EUTRA_DRB_IdentityList was modified so that in case of no AM DRB, v_DRB_IdentityList.AMonly[i] is initialized to omit thus avoiding the execution error. 

	TTCN module
	EUTRA_RRCSteps.ttcn

	MCC160 Comment
	


Before change:

	 /*

   * @desc      local function to get list of all DRBs and list of AM DRBs

   * @param     p_DrbConfigList

   * @return    DRB_IdentityList_AllAndAMonly_Type

   * @status    APPROVED

   */

  function fl_EUTRA_DRB_IdentityList(template (value) RadioBearerList_Type p_DrbConfigList) return DRB_IdentityList_AllAndAMonly_Type

  {

    var DRB_IdentityList_AllAndAMonly_Type v_DRB_IdentityList;

    var DRB_Identity v_DRB_Identity;

    var integer k;

    var integer i := 0;

    for (k:=0; k < lengthof(p_DrbConfigList); k:= k+1) {

      v_DRB_Identity := valueof(p_DrbConfigList[k].Id.Drb);

      v_DRB_IdentityList.All[k] := v_DRB_Identity;

      if (ischosen(p_DrbConfigList[k].Config.AddOrReconfigure.Rlc.Rb.AM)) {

        v_DRB_IdentityList.AMonly[i] := v_DRB_Identity;

        i := i+1;

      }

    }

    return v_DRB_IdentityList;

  }


After change:

	/*

   * @desc      local function to get list of all DRBs and list of AM DRBs

   * @param     p_DrbConfigList

   * @return    DRB_IdentityList_AllAndAMonly_Type

   * @status    APPROVED

   */

  function fl_EUTRA_DRB_IdentityList(template (value) RadioBearerList_Type p_DrbConfigList) return DRB_IdentityList_AllAndAMonly_Type

  {

    var DRB_IdentityList_AllAndAMonly_Type v_DRB_IdentityList;

    var DRB_Identity v_DRB_Identity;

    var integer k;

    var integer i := 0;

    for (k:=0; k < lengthof(p_DrbConfigList); k:= k+1) {

      v_DRB_Identity := valueof(p_DrbConfigList[k].Id.Drb);

      v_DRB_IdentityList.All[k] := v_DRB_Identity;

      if (ischosen(p_DrbConfigList[k].Config.AddOrReconfigure.Rlc.Rb.AM)) {

        v_DRB_IdentityList.AMonly[i] := v_DRB_Identity;

        i := i+1;

      } else {



v_DRB_IdentityList.AMonly[i] := omit;


 }


    }

    return v_DRB_IdentityList;

  }


Change 2 – Correction to f_EUTRA_RRC_CipherActTime_GetForHO
	Test case name
	f_EUTRA_RRC_CipherActTime_GetForHO

	Reason for change
	In test step 4 of fl_EUTRA_7_2_3_21_TestBody there is a call to function f_EUTRA_RRC_CipherActTime_GetForHO. In this function there is a new optional parameter to hand over pdcp counts otherwise obtained from the SS. But due to TTCN implementation, the value is obtained from the SS regardless of this optional parameter being present or not. 

	Summary of change
	In function f_EUTRA_RRC_CipherActTime_GetForHO, duplicated assignment to v_PdcpCountInfoList is deleted after the “if else” statement. 

	TTCN module
	EUTRA_SecuritySteps.ttcn

	MCC160 Comment
	


Before change:

	  function f_EUTRA_RRC_CipherActTime_GetForHO(CellId_Type p_CellId,

                                              DRB_IdentityList_Type p_AM_DrbList,

                                              template (omit) PdcpCountInfoList_Type p_PdcpCountInfoList := omit) runs on EUTRA_PTC return SecurityActTimeList_Type

  { /* Common functon to enquire PDCP sequence numbers and generate a ciphering activation list for SRB1, SRB2 and DRBs with SNs reset to 0 for SRBs and UM DRBs;

     * Note: this is done immediately i.e. without timing info but HO scenarios may require timing info;

     *       therefore between calling this function and reconfiguring ciphering there shall be no data on any of the given AM bearers in UL or DL */

    /* @sic R5-120721 change 7: new optional parameter to hand over PdcpCounts instead of getting it from the SS sic@ */

    var PdcpCountInfoList_Type v_PdcpCountInfoList;

    var PdcpCountInfo_Type v_PdcpCountInfo;

    var SecurityActTime_Type v_SecurityActTime;

    var SecurityActTimeList_Type v_SecurityActTimeList;

    var integer i;

    if (isvalue(p_PdcpCountInfoList)) {

      v_PdcpCountInfoList := valueof(p_PdcpCountInfoList);               // @sic R5-120721 change 7 sic@

    } else {

      v_PdcpCountInfoList := f_EUTRA_SS_PdcpCount_GetAll(p_CellId);

    }

    v_PdcpCountInfoList := f_EUTRA_SS_PdcpCount_GetAll(p_CellId);
    for (i := 0; i < lengthof(v_PdcpCountInfoList); i:= i + 1) {

      v_PdcpCountInfo := v_PdcpCountInfoList[i];

      if (not fl_PdcpCount_CheckAmDrb(v_PdcpCountInfo, p_AM_DrbList)) {

        v_PdcpCountInfo.UL.Value := int2bit(0, 32);

        v_PdcpCountInfo.DL.Value := int2bit(0, 32);

      }

      v_SecurityActTime.RadioBearerId := v_PdcpCountInfo.RadioBearerId;

      v_SecurityActTime.UL.SQN := fl_EUTRA_SQN_ConvertFromPdcpCount(v_PdcpCountInfo.UL);

      v_SecurityActTime.DL.SQN := fl_EUTRA_SQN_ConvertFromPdcpCount(v_PdcpCountInfo.DL);

      v_SecurityActTimeList[i] := v_SecurityActTime;

    }

    return v_SecurityActTimeList;

  }


After change:

	 function f_EUTRA_RRC_CipherActTime_GetForHO(CellId_Type p_CellId,

                                              DRB_IdentityList_Type p_AM_DrbList,

                                              template (omit) PdcpCountInfoList_Type p_PdcpCountInfoList := omit) runs on EUTRA_PTC return SecurityActTimeList_Type

  { /* Common functon to enquire PDCP sequence numbers and generate a ciphering activation list for SRB1, SRB2 and DRBs with SNs reset to 0 for SRBs and UM DRBs;

     * Note: this is done immediately i.e. without timing info but HO scenarios may require timing info;

     *       therefore between calling this function and reconfiguring ciphering there shall be no data on any of the given AM bearers in UL or DL */

    /* @sic R5-120721 change 7: new optional parameter to hand over PdcpCounts instead of getting it from the SS sic@ */

    var PdcpCountInfoList_Type v_PdcpCountInfoList;

    var PdcpCountInfo_Type v_PdcpCountInfo;

    var SecurityActTime_Type v_SecurityActTime;

    var SecurityActTimeList_Type v_SecurityActTimeList;

    var integer i;

    if (isvalue(p_PdcpCountInfoList)) {

      v_PdcpCountInfoList := valueof(p_PdcpCountInfoList);               // @sic R5-120721 change 7 sic@

    } else {

      v_PdcpCountInfoList := f_EUTRA_SS_PdcpCount_GetAll(p_CellId);

    }

    //v_PdcpCountInfoList := f_EUTRA_SS_PdcpCount_GetAll(p_CellId);
    for (i := 0; i < lengthof(v_PdcpCountInfoList); i:= i + 1) {

      v_PdcpCountInfo := v_PdcpCountInfoList[i];

      if (not fl_PdcpCount_CheckAmDrb(v_PdcpCountInfo, p_AM_DrbList)) {

        v_PdcpCountInfo.UL.Value := int2bit(0, 32);

        v_PdcpCountInfo.DL.Value := int2bit(0, 32);

      }

      v_SecurityActTime.RadioBearerId := v_PdcpCountInfo.RadioBearerId;

      v_SecurityActTime.UL.SQN := fl_EUTRA_SQN_ConvertFromPdcpCount(v_PdcpCountInfo.UL);

      v_SecurityActTime.DL.SQN := fl_EUTRA_SQN_ConvertFromPdcpCount(v_PdcpCountInfo.DL);

      v_SecurityActTimeList[i] := v_SecurityActTime;

    }

    return v_SecurityActTimeList;

  }

